

## Applicazioni numeriche per lo studio delle direttività

### 1) Generalità

Lo studio delle caratteristiche di direttività delle basi idrofoniche riceventi e trasmettenti gioca un ruolo fondamentale nella progettazione dei sistemi sonar; le forme geometriche di questi manufatti sono le più diverse in dipendenza del tipo di piattaforma che le deve alloggiare e della funzione che devono svolgere.

Non sempre è possibile disporre di algoritmi di calcolo che si adattino esattamente alla forma della base, si deve in questi casi impiegare il metodo di calcolo che più si avvicina al modello fisico da studiare accettando gli inevitabili scostamenti dal vero che questa procedura può portare; in questi casi però le indicazioni fornite dai calcoli sono comunque sufficienti a consentire una progettazione corretta.

### 2) Applicazioni numeriche per le basi lineari ad n idrofoni

La prima applicazione numerica che ci accingiamo a svolgere è relativa al calcolo delle direttività "R" di basi rettilinee di lunghezza "L", costituite da "n" idrofoni disposti ad eguale distanza tra loro disposti per ricevere segnale in banda f1-f2.

L'algoritmo che impieghiamo è dovuto a STENZEL ed è così strutturato:

$$R = \left\{ \left( \frac{1}{n} \right) + \left( \frac{2}{n^2} \right) \sum_{m=1}^{n-1} (n-m) [\text{Sen}(mpx)/(mpx)] \text{Cos} [(p+2)mx] \right\}^{1/2}$$

n= numero degli idrofoni

L=lunghezza della base in metri

dove  $x = (\pi d/\lambda) \text{Sen } \gamma$

$d = L/(n-1)$

$\gamma$ =angolo rispetto all'asse della base

f1= frequenza inferiore della banda

f2=frequenza superiore della banda

$\lambda = c/f2 = 1530/f2$

$p = (f2-f1)/f1$

Il calcolo secondo l'algoritmo illustrato ed il tracciamento della curva di direttività risultante si esegue con facilità soltanto se si impiega il computer; per tale ragione è stato impostato il seguente programma in Qbasic:

```
'CALCOLO CURVE DI DIRETTIVITA' BASE LINEARE SECONDO STENZEL
```

```
'Campo angolare esplorato da 0° ad 80°
```

```
'L=lunghezza della base in metri
```

```
'n=numero degli idrofoni
```

```
'f1;f2 frequenze che limitano la banda
```

```
ini:
```

```
CLS
```

```
SCREEN 9
```

```
FOR x = 0 TO 460 STEP 23
```

```
FOR y = 0 TO 320 STEP 2
```

```
PSET (x, y), 7
```

```
NEXT y
```

```
NEXT x
```

```
FOR y = 0 TO 320 STEP 16
```

```
FOR x = 0 TO 460 STEP 3
```

```
PSET (x, y), 7
```

```
NEXT x
```

```
NEXT y
```

```
LINE (0, 160)-(460, 160)
```

```
LINE (0, 0)-(0, 320)
```

```
LOCATE 1, 60: PRINT "DIRETTIVITA'"
```

```
LOCATE 2, 60: PRINT "SECONDO STENZEL"
```

```
LOCATE 4, 60: INPUT "n° idrofoni"; n
```

```
LOCATE 5, 60: INPUT "lung.base mt"; L
```

```

LOCATE 7, 66: INPUT "F1"; f1'frequenza inferiore della banda
LOCATE 8, 66: INPUT "F2"; f2'frequenza superiore della banda

LOCATE 18, 60: PRINT "asse x = 4ø/div."
LOCATE 19, 60: PRINT "asse y = .1/div."

d1 = L / (n - 1)'calcolo della distanza media tra gli n idrofoni
p = (f2 - f1) / f1 'coeff. richiesto da formula STENZEL
FOR a = 0 TO 80 STEP .1 'gradi sessagesimali di calcolo
x = 3.14 * d1 * (f2 / 1530) * SIN((a + .000001) * (3.14 / 180))
FOR M = 1 TO (n - 1)
b = (SIN(M * p * x)) / (M * p * x)
c = COS((p + 2) * M * x)
d = (n - M)
E = (b * c * d)
K = K + E
NEXT M
s = ((2 / (n ^ 2)) * K) + (1 / n)
t = SQR(s)
K = 0
PSET ((a * 460 / 80), (160 - 160 * t)), 7
NEXT a

fine:
a$ = INKEY$
IF (a$) = CHR$(27) THEN END
IF (a$) = CHR$(114) THEN GOTO ini
GOTO fine

```

Impieghiamo ora il programma illustrato per il calcolo della direttività di una base rettilinea che deve operare con i seguenti parametri:

numero degli idrofoni  $n=9$

lunghezza della base  $L=1$  metro

frequenza inferiore della banda  $f_1=1500\text{Hz}$

frequenza superiore della banda  $f_2=3000\text{Hz}$

inserite in successione le variabili nel programma che gira su P.C. si ottiene la curva di direttività mostrata in figura 1




Figura 1

Dalla figura si vede come la semilarghezza del lobo principale a  $-3\text{dB}$  è ampia  $16^\circ$  pari ad una larghezza totale del lobo di  $32^\circ$ .

Sempre dalla figura si osserva come i lobi secondari ondolino sotto le due divisioni con una attenuazione dell'ordine di circa  $16\text{ dB}$  rispetto al lobo principale.

Se si ripete il calcolo per una banda di frequenze posta più in alto nello spettro con  $f_1=8000\text{ Hz}$  e  $f_2=16000\text{ Hz}$  si ottiene la caratteristica di direttività tracciata in figura2




Figura 2

Dalla figura si vede come la semilarghezza del lobo principale a  $-3\text{dB}$  è ampia  $4^\circ$  pari ad una larghezza totale del lobo di  $8^\circ$ .

Sempre dalla figura si osserva come i lobi secondari ondolino sotto le tre divisioni con una attenuazione dell'ordine di circa  $13\text{ dB}$  rispetto al lobo principale.

### 3) Applicazioni numeriche per le basi lineari continue

La seconda applicazione numerica che ci accingiamo a svolgere è relativa al calcolo delle direttività "R" di piccole basi rettilinee di lunghezza "l", costituite da un unico segmento sensibile disposto a ricevere il segnale alla frequenza "f".

L'algoritmo che impieghiamo è così strutturato:

$$R = \frac{\text{Sen}[(\pi l/\lambda) \text{Sen}\beta]}{[(\pi l/\lambda) \text{Sen}\beta]}$$

Dove

l= lunghezza del segmento

$$\lambda = c/f$$

$\beta$ = angolo rispetto all'asse della linea

Si tratta di una funzione di direttività più semplice della precedente ma che necessita sempre l'impiego di un personal computer per il calcolo e il tracciamento del grafico; il programma di seguito riportato è stato preparato allo scopo:

```
'CALCOLO CURVE DI DIRETTIVITA' BASE LINEARE UNIFORME
```

```
'l=lunghezza della base in metri
```

```
'f= frequenza di lavoro
```

```
ini:
```

```
CLS
```

```
SCREEN 9
```

```
FOR x = 20 TO 480 STEP 23
```

```
FOR y = 10 TO 330 STEP 2
```

```
PSET (x, y), 7
```

```
NEXT y
```

```
NEXT x
```

```
FOR y = 10 TO 330 STEP 16
```

```
FOR x = 20 TO 480 STEP 3
```

```
PSET (x, y), 7
```

```
NEXT x
```

```
NEXT y
```

```
LINE (20, 170)-(480, 170)
```

```
LINE (20, 10)-(20, 330)
```

```
LOCATE 1, 62: PRINT "DIRETTIVITA'"
```

```
LOCATE 2, 62: PRINT "BASE LINEARE"
```

```
LOCATE 3, 62: PRINT " UNIFORME"
```

```
LOCATE 5, 62: INPUT "lung.base mt"; l
```

```
LOCATE 7, 66: INPUT "F"; f'frequenza di lavoro
```

```
LOCATE 18, 60: PRINT "asse x = 4ø/div."
```

```

LOCATE 19, 60: PRINT "asse y = .1/div."

FOR a = 0 TO 80 STEP .1 'gradi sessagesimali di calcolo
x = 3.14 * l * (f / 1530) * SIN((a + .000001) * (3.14 / 180))
r = (SIN(x)) / (x)
PSET ((20 + a * 460 / 80), (170 - 160 * r)), 7
NEXT a

fine:

a$ = INKEY$

IF (a$) = CHR$(27) THEN END

IF (a$) = CHR$(114) THEN GOTO ini

GOTO fine

```


Impieghiamo ora il programma illustrato per il calcolo della direttività di una base rettilinea uniforme che deve operare con i seguenti parametri:

lunghezza della base  $l = 0.1$  metri

frequenza di lavoro  $f = 50000$  Hz

Dalla figura 3 si vede come la semilarghezza del lobo principale a  $-3$ dB è ampia  $8^\circ$  pari ad una larghezza totale del lobo di  $16^\circ$ .

Sempre dalla figura si osserva come i lobi secondari ondolino sotto le tre divisioni con una attenuazione dell'ordine di circa 13 dB rispetto al lobo principale.



#### 4) Applicazioni numeriche per le basi circolari ad n idrofoni

La terza applicazione numerica che ci accingiamo a svolgere è relativa al calcolo delle direttività “R” di basi a forma circolare di diametro “D” con “n” idrofoni operanti alla frequenza “F”.

La caratteristica di direttività in oggetto è espressa secondo l’algoritmo

$$R = J_0 \left[ \left( \frac{2\pi D}{\lambda} \right) \text{Sen} \left( \frac{\alpha}{2} \right) \right]$$

Dove  $J_0$  è il simbolo della funzione di Bessel di ordine zero

$$\lambda = c/f$$

$\alpha$  = angolo rispetto al diametro considerato l’asse della base

ed è valida soltanto se  $n > \left( \frac{2\pi D}{\lambda} \right) + 2$

L’algoritmo è classico ed è riportato in tutti i testi che trattano di acustica subaquea; l’unica difficoltà di calcolo consiste nell’utilizzo delle funzioni di Bessel delle quali sono disponibili ottime tavole che non consentono però di calcolare e tracciare in modo continuo la curva di direttività; grazie all’aiuto del programma che illustreremo il calcolo della  $J_0$  si esegue mediante la computazione, a carico del programma, dell’integrale:

$$J_0 = \int_0^{\pi} [\text{Cos}(z \text{ Cos } q)] dq$$

'DIRETTIVITA' DI UNA BASE CIRCOLARE

'D=diametro del cerchio

'f= frequenza di lavoro

ini:

CLS

SCREEN 9

FOR x = 20 TO 480 STEP 23

FOR y = 10 TO 330 STEP 2

PSET (x, y), 7

NEXT y

NEXT x

FOR y = 10 TO 330 STEP 16

FOR x = 20 TO 480 STEP 3

PSET (x, y), 7

NEXT x

NEXT y

LINE (20, 170)-(480, 170)

LINE (20, 10)-(20, 330)

LOCATE 1, 62: PRINT "DIRETTIVITA'"

LOCATE 2, 62: PRINT "BASE CIRCOLARE"

LOCATE 5, 62: INPUT "diametro mt"; d

LOCATE 7, 66: INPUT "F"; f'frequenza di lavoro

LOCATE 18, 60: PRINT "asse x = 4ø/div."

```

LOCATE 19, 60: PRINT "asse y = .1/div."
FOR a = 0 TO 80 STEP .1 'gradi sessagesimali di calcolo
'CALCOLO DELLA DIRETTIVITA' CON LA FUNZIONE DI BESSEL Jo
landa = 1530 / f
z = (6.28 * d / landa) * SIN((a / 2) * 3.14 / 180)
'CALCOLO DELL'INTEGRALE DEFINITO PER LA
'FUNZIONE DI BESSEL ORDINE ZERO--> Jo
a1 = 0
b1 = 3.14
s1 = .001
I1 = 0
FOR x1 = a1 TO b1 STEP s1
y1 = (1 / 3.14) * COS(z * COS(x1))
p1 = y1 * s1
I1 = p1 + I1
NEXT x1
PSET ((20 + a * 460 / 80), (170 - 160 * I1)), 15
NEXT a
fine:
a$ = INKEY$
IF (a$) = CHR$(27) THEN END
IF (a$) = CHR$(114) THEN GOTO ini
GOTO fine

```

Impieghiamo ora il programma illustrato per il calcolo della direttività di una base circolare che deve operare con i seguenti parametri:

diametro base  $D = 1.5$  metri

frequenza di lavoro  $f=10000$  Hz

numero degli idrofoni  $n = 18$

prima di iniziare il calcolo dobbiamo verificare che sia  $n > (2\pi D/\lambda) + 2$  cioè

$$6.28*1.5/(1530/10000)+2 = 3.4 < 18 \text{ e così è}$$

Inserendo ora nel P.C. i valori del diametro  $D$  e della frequenza  $f$  si ottiene la curva di direttività riportata in figura 4; si osserverà che, anche con un P.C. veloce, la curva prenderà corpo poco alla volta dato il notevole carico di calcolo costituito dallo sviluppo dell'integrale per ciascun punto di calcolo.




Figura 4

Dalla figura 4 si vede come la semilarghezza del lobo principale a  $-3$ dB è ampia circa  $2^\circ$  pari ad una larghezza totale del lobo di  $4^\circ$ .

Sempre dalla figura si osserva come i lobi secondari ondolino sotto le quattro divisioni con una attenuazione dell'ordine di circa 12 dB rispetto al lobo principale.

##### 5) Applicazioni numeriche per le basi lineari in correlazione

La quinta applicazione numerica che ci accingiamo a svolgere è relativa al calcolo delle direttività "R" di basi lineari ad "n" idrofoni (con n pari) che formano la loro caratteristica a seguito di un processo multiplo di correlazione.

In questo caso le coppie di idrofoni simmetriche, rispetto all'asse della base, vengono correlate tra loro e il risultato di questa operazione viene sommato alle analoghe

operazioni di correlazione eseguite tra le altre coppie; se ad esempio abbiamo 2 coppie di idrofoni si potranno correlare tra loro 2 coppie e dalla somma di queste 2 correlazioni trarre la caratteristica di direttività della base.

L'algoritmo impiegato per questo calcolo è il seguente:

$$R_c = R_1 + R_2$$

$$R_1 = [\text{Sen}(2\pi D_f x_1) \text{Cos}(2\pi F_o x_1)] / (2\pi D_f x_1)$$

$$R_2 = [\text{Sen}(2\pi D_f x_2) \text{Cos}(2\pi F_o x_2)] / (2\pi D_f x_2)$$

Dove  $D_f = (f_2 - f_1) / 2$

$$F_o = (f_1 + f_2) / 2$$

$f_1$  = frequenza inferiore della banda

$f_2$  = frequenza superiore della banda

$$x_1 = (d_1 / c) \text{Sen } \alpha$$

$$x_2 = (d_2 / c) \text{Sen } \alpha$$

e  $d_1, d_2$  sono le distanze tra gli idrofoni di ciascuna coppia ed  $\alpha$  è la direzione di scostamento dall'asse della base

Sempre con l'ausilio del personal computer implementiamo il seguente programma di calcolo per la funzione  $R_c$ :

```
'DIRETTIVITA' DI UNA BASE LINEARE A 4 IDROFONI IN CORRELAZIONE
```

```
ini:
```

```
CLS
```

```
SCREEN 9
```

```
FOR x = 20 TO 480 STEP 23
```

```
FOR y = 10 TO 330 STEP 2
```

```
PSET (x, y), 7
```

```
NEXT y
```

```
NEXT x
```

```
FOR y = 10 TO 330 STEP 16
```

```
FOR x = 20 TO 480 STEP 3
```

```

PSET (x, y), 7

NEXT x

NEXT y

LINE (20, 170)-(480, 170)

LINE (20, 10)-(20, 330)

LOCATE 1, 62: PRINT "DIRETTIVITA'"
LOCATE 2, 62: PRINT "IN CORRELAZIONE"

LOCATE 4, 66: INPUT "F1"; f1'frequenza inferiore

LOCATE 5, 66: INPUT "F2"; f2'frequenza superiore

LOCATE 6, 66: INPUT "d1"; d1'distanza tra idrofoni prima coppia

LOCATE 7, 66: INPUT "d2"; d2'distanza tra idrofoni seconda coppia

DF = (f2 - f1) / 2

Fo = (f2 + f1) / 2

FOR a = .001 TO 80 STEP .05
'calcolo variabili

x = 6.28 * DF * SIN(3.14 * a / 180) / 1530

xo = 6.28 * Fo * SIN(3.14 * a / 180) / 1530

'FUNZIONE DI CORRELAZIONE ANALOGICA

Cb = .5*((SIN(x*d1)/(x*d1))*COS(xo*d1)) + (SIN(x*d2)/(x*d2))* COS(xo*d2))

PSET ((20 + a * 460 / 80), (170 - 160 * Cb))'GRAFICO di Cb

NEXT
fine:

a$ = INKEY$

IF (a$) = CHR$(27) THEN END

IF (a$) = CHR$(114) THEN GOTO ini

GOTO fine

```

Impieghiamo ora il programma illustrato per il calcolo della direttività di una base a 4 idrofoni in correlazione che deve operare con i seguenti parametri:

distanza tra gli idrofoni della prima coppia  $d1 = 1$  metro

distanza tra gli idrofoni della seconda coppia  $d2 = 0.5$  metri

frequenza inferiore della banda  $f1 = 1000$  Hz

frequenza superiore della banda  $f2 = 10000$  Hz




Figura 5

Dalla figura 5 si vede come la semilarghezza del lobo principale a  $-3$ dB è ampia circa  $2.5^\circ$  pari ad una larghezza totale del lobo di  $5^\circ$ .

Sempre dalla figura si osserva come i lobi secondari ondolino sotto le tre divisioni con una attenuazione dell'ordine di circa 13 dB rispetto al lobo principale.